A Hierarchy of Authentication Properties

Lowe's hierarchy

- In the literature, you will see multiple variations of authentication claims with subtle differences
- In 1997, Gavin Lowe proposed **a hierarchy of authentication properties** in increasing strength:
  1. aliveness,
  2. weak agreement,
  3. non-injective agreement, and
  4. injective agreement
- Each property applies from either party's perspective

Authentication claims

We typically use the following action facts:

- `Create(X)` – Create Agent X
- `Running(X,Y,t)` – Agent X believes to be executing the protocol with agent Y and has learned term t
- `Commit(X,Y,t)` – Agent X believes to have completed the protocol with agent Y and has learned term t
- `Compromised(X)` – Agent X has compromised its long-term secret(s)



Aliveness

A protocol guarantees **aliveness**
to an agent *a* in role *A* if, whenever *a* completes a run of the protocol, seemingly with *b* in role *B*, then *b* has previously been running the protocol.

```
/* Aliveness from A's perspective */
lemma aliveness:
    " All A B t #i .
        Commit(A, B, t)@i
        & not (Ex #r. Compromised(A)@r)
        & not (Ex #r. Compromised(B)@r)
            ==> (Ex #j. Create(B)@j & j < i)
    "
```

Weak agreement

A protocol guarantees **weak agreement**
to an agent *a* in role *A* if, whenever *a* completes a run of the protocol, seemingly with *b* in role *B*, then *b* has previously been running the protocol, seemingly with *a*.

```
/* Weak agreement from A's perspective */
lemma weak_agreement:
    " All A B t #i .
        Commit(A, B, t)@i
        & not (Ex #r. Compromised(A)@r)
        & not (Ex #r. Compromised(B)@r)
            ==> (Ex t2 #j. Running(B, A, t2)@j & j < i)
    "
```

Non-injective agreement

A protocol guarantees **non-injective agreement** to an agent *a* in role *A* if, whenever *a* completes a run of the protocol, seemingly with *b* in role *B*, then *b* has previously been running the protocol in role *B*, seemingly with *a*, and they both agree on the term *t*.

```
/* Non-injective agreement from A's perspective */
lemma noninjective_agreement:
    " All A B t #i .
        Commit(A, B, t) @i
        & not (Ex #r. Compromised(A)@r)
        & not (Ex #r. Compromised(B)@r)
            ==> (Ex #j. Running(B, A, t)@j & j < i)
    "
```

Injective agreement

A protocol guarantees **injective agreement**
to an agent *a* in role *A* if, whenever *a* completes a run of the protocol, seemingly with *b* in role *B*, then *b* has previously been running the protocol in role *B*, seemingly with *a*, and they both agree on the term *t*.
Each run of agent *a* in role *A* corresponds to a unique run of agent *b*.

```
/* Injective agreement from A's perspective */
lemma injective_agreement:
    " All A B t #i.
        Commit(A, B, t)@i
        & not (Ex #r. Compromised(A)@r)
        & not (Ex #r. Compromised(B)@r)
            ==> (Ex #j. Running(B, A, t)@j & j < i
                  & not(Ex A2 B2 #i2 . Commit(A2, B2, t)@i2
                     & not(#i = #i2)))
    "
```